
Il seguente esempio del componente può essere utilizzato così com'è cambiando solo gli indirizzi email. 
Il server smtp è già impostato per funzionare correttamente, nei campi .to .cc .bcc è possibile specificare più destinatari semplicemente separando gli indirizzi email con il carattere ; nell'esempio si veda il campo .cc che è impostato 
per inviare l'email a due destinatari. Se l'email è in forma testuale si deve utilizzare per il corpo del messaggio la propietà .TextBody, se invece il messaggio è in formato HTML si assegna il corpo del messaggio alla propietà .HtmlBody 


<!--METADATA TYPE="typelib" UUID="CD000000-8B95-11D1-82DB-00C04FB1625D" NAME="CDO for Windows 2000 Library" --> 
<!--METADATA TYPE="typelib" UUID="00000205-0000-0010-8000-00AA006D2EA4" NAME="ADODB Type Library" --> 

<html> 
<body> 

<% 
Dim objMessage, objConfig 

'Crea gli oggetti per il messaggio e per la configurazione 
Set objMessage = CreateObject("CDO.Message") 
Set objConfig = CreateObject("CDO.Configuration") 

'Configurazione server smtp 
objConfig.Fields(cdoSendUsingMethod) = cdoSendUsingPort 
objConfig.Fields(cdoSMTPServer) = "localhost" 
objConfig.Fields(cdoSMTPServerPort) = 25 

'Aggiornamento configurazione 
objConfig.Fields.Update 
Set objMessage.Configuration = objConfig 

'crea il messaggio 
objMessage.To = " info@nomedominio.xx " 
objMessage.Cc = " test@destinatariocc.xx ; test@destinatariocc2.xx " 
objMessage.Bcc = " test@destinatariobcc.xx " 
objMessage.From = """NT"" < clienti@consultingweb.it >" 
objMessage.ReplyTo = " clienti@consultingweb.it " 
objMessage.Subject = "Invio Email CDOsys" 
'objMessage.TextBody = "messaggio in formato testuale" 
objMessage.HtmlBody = "<html><body>messaggio in formato HTML</body></html>" 

' gestione errori 
On Error Resume Next 

' Send the CDOSYS Message 
objMessage.Send 
Set objMessage = Nothing 
Set objConfig = Nothing 



'visualizza un messaggio di ok in caso di invio senza problemi 
'altrimenti visualizza il messaggio di errore 
If Err.Number = 0 Then 
Response.Write "Il messaggio è stato inviato; nessun errore!<hr>" 
Else 
Response.Write Err.Description & "<hr>" 
Err.Number = 0 
End If 
On Error Goto 0 

%> 
</body> 
</html> 

